NAG Fortran Library Routine Document

F08PPF (ZGEESX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08PPF (ZGEESX) computes the eigenvalues, the Schur form T, and, optionally, the matrix of Schur vectors Z for an n by n complex nonsymmetric matrix A.

2 Specification

```
 SUBROUTINE F08PPF (JOBVS, SORT, SELECT, SENSE, N, A, LDA, SDIM, W, VS,
1                   LDVS, RCONDE, RCONDV, WORK, LWORK, RWORK, BWORK,
2                   INFO)
 INTEGER            N, LDA, SDIM, LDVS, LWORK, INFO
 double precision   RCONDE, RCONDV, RWORK(*)
 complex*16         A(LDA,*), W(*), VS(LDVS,*), WORK(*)
 LOGICAL            SELECT, BWORK(*)
 CHARACTER*1        JOBVS, SORT, SENSE
 EXTERNAL           SELECT
```

The routine may be called by its LAPACK name ***zgeesx***.

3 Description

The Schur factorization of A is given by

$$A = ZTZ^H,$$

where Z is orthogonal, the matrix of Schur vectors, and T is upper triangular.

Optionally, F08PPF (ZGEESX) also orders the eigenvalues on the diagonal of the Schur form so that selected eigenvalues are at the top left; computes a reciprocal condition number for the average of the selected eigenvalues (RCONDE); and computes a reciprocal condition number for the right invariant subspace corresponding to the selected eigenvalues (RCONDV). The leading columns of Z form an orthonormal basis for this invariant subspace.

For further explanation of the reciprocal condition numbers RCONDE and RCONDV, see Section 4.8 of Anderson *et al.* (1999) (where these quantities are called s and sep respectively).

A complex matrix is in Schur form if it is upper triangular.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOBVS – CHARACTER*1 *Input*

On entry: if JOBVS = 'N', Schur vectors are not computed.

If JOBVS = 'V', Schur vectors are computed.

2: SORT – CHARACTER*1 *Input*

On entry: specifies whether or not to order the eigenvalues on the diagonal of the Schur form:

if SORT = 'N', eigenvalues are not ordered;
if SORT = 'S', eigenvalues are ordered (see SELECT).

3: SELECT – LOGICAL FUNCTION, supplied by the user. *External Procedure*

If SORT = 'S', SELECT is used to select eigenvalues to sort to the top left of the Schur form.

If SORT = 'N', SELECT is not referenced and F08PPF (ZGEESX) may be called with the dummy function F08PNZ.

Its specification is:

```
LOGICAL FUNCTION SELECT (W)
complex*16              W
```

1: W – ***complex*16*** *Input*

On entry: the eigenvalue $\mathrm{W}(j)$ is selected if $\mathrm{SELECT}(\mathrm{W}(j))$ is .TRUE..

SELECT must be declared as EXTERNAL in the (sub)program from which F08PPF (ZGEESX) is called. Parameters denoted as *Input* must **not** be changed by this procedure.

4: SENSE – CHARACTER*1 *Input*

On entry: determines which reciprocal condition numbers are computed:

if SENSE = 'N', none are computed;
if SENSE = 'E', computed for average of selected eigenvalues only;
if SENSE = 'V', computed for selected right invariant subspace only;
if SENSE = 'B', computed for both.

If SENSE = 'E', 'V' or 'B', SORT must equal 'S'.

5: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

6: A(LDA,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1,\mathrm{N})$.

On entry: the n by n matrix A.

On exit: is overwritten by its Schur form T.

7: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08PPF (ZGEESX) is called.

Constraint: $\mathrm{LDA} \geq \max(1,\mathrm{N})$.

8: SDIM – INTEGER *Output*

On exit: if SORT = 'N', SDIM = 0.

If SORT = 'S', SDIM = number of eigenvalues for which SELECT is true.

9: W($*$) – ***complex*16*** array *Output*

Note: the dimension of the array W must be at least $\max(1, \mathrm{N})$.

On exit: contains the computed eigenvalues, in the same order that they appear on the diagonal of the output Schur form T.

10: VS(LDVS,$*$) – ***complex*16*** array *Output*

Note: the second dimension of the array VS must be at least $\max(1, \mathrm{N})$ if JOBVS = 'V' and at least 1 otherwise.

On exit: if JOBVS = 'V', VS contains the unitary matrix Z of Schur vectors.

If JOBVS = 'N', VS is not referenced.

11: LDVS – INTEGER *Input*

On entry: the first dimension of the array VS as declared in the (sub)program from which F08PPF (ZGEESX) is called.

Constraints:

if JOBVS = 'V', LDVS $\geq \max(1, \mathrm{N})$;
LDVS ≥ 1 otherwise.

12: RCONDE – ***double precision*** *Output*

On exit: if SENSE = 'E' or 'B', contains the reciprocal condition number for the average of the selected eigenvalues.

If SENSE = 'N' or 'V', RCONDE is not referenced.

13: RCONDV – ***double precision*** *Output*

On exit: if SENSE = 'V' or 'B', RCONDV contains the reciprocal condition number for the selected right invariant subspace.

If SENSE = 'N' or 'E', RCONDV is not referenced .

14: WORK($*$) – ***complex*16*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if INFO $= 0$, $\mathrm{WORK}(1)$ returns the optimal LWORK.

15: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08PPF (ZGEESX) is called.

For good performance, LWORK must generally be larger than the minimum; increase the workspace by, say, $nb \times \mathrm{N}$, where nb is the optimal block size for F08NSF (ZGEHRD).

If LWORK $= -1$, a workspace query is assumed; the routine only calculates an upper bound on the optimal size of the array WORK, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

If SENSE = 'E', 'V' or 'B', LWORK $\geq 2 \times \mathrm{SDIM} \times (\mathrm{N} - \mathrm{SDIM})$, where SDIM is the number of selected eigenvalues computed by this routine.

Note that $2 \times \mathrm{SDIM} \times (\mathrm{N} - \mathrm{SDIM}) \leq \mathrm{N} \times \mathrm{N}/2$. Note also that an error is only returned if $\mathrm{LWORK} < \max(1, 2 \times \mathrm{N})$, but if SENSE = 'E', 'V' or 'B' this may not be large enough.

Constraint: $\mathrm{LWORK} \geq \max(1, 2 \times \mathrm{N})$.

16: RWORK($*$) – ***double precision*** array *Workspace*

Note: the dimension of the array RWORK must be at least $\max(1, \mathrm{N})$.

17: BWORK($*$) – LOGICAL array *Workspace*

Note: the dimension of the array BWORK must be at least 1 if SORT = 'N' and at least $\max(1, \mathrm{N})$ otherwise.

If SORT = 'N', BWORK is not referenced.

18: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If $\mathrm{INFO} = -i$, the ith argument had an illegal value.

INFO > 0

If $\mathrm{INFO} = i$ and $i \leq \mathrm{N}$, the QR algorithm failed to compute all the eigenvalues

INFO = N + 1

The eigenvalues could not be reordered because some eigenvalues were too close to separate (the problem is very ill-conditioned).

INFO = N + 2

After reordering, roundoff changed values of some complex eigenvalues so that leading eigenvalues in the Schur form no longer satisfy SELECT = .TRUE.. This could also be caused by underflow due to scaling.

7 Accuracy

The computed Schur factorization satisfies

$$A + E = ZTZ^T,$$

where

$$\|E\|_2 = \mathrm{O}(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***. See Section 4.8 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating-point operations is proportional to n^3.

The real analogue of this routine is F08PBF (DGEESX).

9 Example

To find the Schur factorization of the matrix

$$A = \begin{pmatrix} -3.97 - 5.04i & -4.11 + 3.70i & -0.34 + 1.01i & 1.29 - 0.86i \\ 0.34 - 1.50i & 1.52 - 0.43i & 1.88 - 5.38i & 3.36 + 0.65i \\ 3.31 - 3.85i & 2.50 + 3.45i & 0.88 - 1.08i & 0.64 - 1.48i \\ -1.10 + 0.82i & 1.81 - 1.59i & 3.25 + 1.33i & 1.57 - 3.44i \end{pmatrix},$$

such that the eigenvalues A with positive real part of are the top left diagonal elements of the Schur form, T. Estimates of the condition numbers for the selected eigenvalue cluster and corresponding invariant subspace are also returned.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08PPF Example Program Text
*     Mark 21.  NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NB, NMAX
      PARAMETER        (NB=64,NMAX=10)
      INTEGER          LDA, LDVS, LWORK
      PARAMETER        (LDA=NMAX,LDVS=NMAX,LWORK=NMAX*(NB+1+NMAX/2))
*     .. Local Scalars ..
      DOUBLE PRECISION ANORM, EPS, RCONDE, RCONDV, TOL
      INTEGER          I, IFAIL, INFO, J, LWKOPT, N, SDIM
*     .. Local Arrays ..
      COMPLEX *16      A(LDA,NMAX), VS(LDVS,NMAX), W(NMAX), WORK(LWORK)
      DOUBLE PRECISION RWORK(NMAX)
      LOGICAL          BWORK(NMAX)
*     .. External Functions ..
      DOUBLE PRECISION F06UAF, X02AJF
      LOGICAL          SELECT
      EXTERNAL         F06UAF, X02AJF, SELECT
*     .. External Subroutines ..
      EXTERNAL         X04DAF, ZGEESX
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08PPF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read in the matrix A
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
*
*        Find the Frobenius norms of A
*
         ANORM = F06UAF('Frobenius',N,N,A,LDA,RWORK)
*
*        Find the Schur factorization of A
*
         CALL ZGEESX('Vectors (Schur)','Sort',SELECT,
     +               'Both reciprocal condition numbers',N,A,LDA,SDIM,W,
     +               VS,LDVS,RCONDE,RCONDV,WORK,LWORK,RWORK,BWORK,INFO)
         LWKOPT = WORK(1)
*
         IF (INFO.EQ.0 .OR. INFO.EQ.(N+2)) THEN
*
```

```
*           Print solution
*
            WRITE (NOUT,99999)
     +        'Number of eigenvalues for which SELECT is true = ', SDIM,
     +        '(dimension of invariant subspace)'
            WRITE (NOUT,*)
            IF (INFO.EQ.(N+2)) THEN
               WRITE (NOUT,99998) '***Note that rounding errors mean ',
     +           'that leading eigenvalues in the Schur form',
     +           'no longer satisfy SELECT = .TRUE.'
               WRITE (NOUT,*)
            END IF
*
*           Print out factors of the Schur factorization
*
            IFAIL = 0
            CALL X04DAF('General',' ',N,N,A,LDA,'Schur matrix T',IFAIL)
*
            WRITE (NOUT,*)
            CALL X04DAF('General',' ',N,N,VS,LDVS,
     +                  'Matrix of Schur vectors Z',IFAIL)
*
*           Print out the reciprocal condition numbers
*
            WRITE (NOUT,*)
            WRITE (NOUT,99997)
     +        'Reciprocal of projection norm onto the invariant',
     +        'subspace for the selected eigenvalues', 'RCONDE = ',
     +        RCONDE
            WRITE (NOUT,*)
            WRITE (NOUT,99996)
     +        'Reciprocal condition number for the invariant subspace',
     +        'RCONDV = ', RCONDV
*
*           Compute the machine precision
*
            EPS = X02AJF()
            TOL = EPS*ANORM
*
*           Print out the approximate asymptotic error bound on the
*           average absolute error of the selected eigenvalues given by
*
*              eps*norm(A)/RCONDE
*
            WRITE (NOUT,*)
            WRITE (NOUT,99995)
     +        'Approximate asymptotic error bound for selected ',
     +        'eigenvalues   = ', TOL/RCONDE
*
*           Print out an approximate asymptotic bound on the maximum
*           angular error in the computed invariant subspace given by
*
*              eps*norm(A)/RCONDV
*
            WRITE (NOUT,99995)
     +        'Approximate asymptotic error bound for the invariant ',
     +        'subspace = ', TOL/RCONDV
         ELSE
            WRITE (NOUT,99994) 'Failure in ZGEESX. INFO =', INFO
         END IF
*
*        Print workspace information
*
         IF (LWORK.LT.LWKOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99993) 'Optimum workspace required = ', LWKOPT,
     +        'Workspace provided         = ', LWORK
         END IF
      ELSE
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'NMAX too small'
```

```
      END IF
      STOP
*
99999 FORMAT (1X,A,I4,/1X,A)
99998 FORMAT (1X,2A,/1X,A)
99997 FORMAT (1X,A,/1X,A,/1X,A,1P,E8.1)
99996 FORMAT (1X,A,/1X,A,1P,E8.1)
99995 FORMAT (1X,2A,1P,E8.1)
99994 FORMAT (1X,A,I4)
99993 FORMAT (1X,A,I5,/1X,A,I5)
      END

      LOGICAL FUNCTION SELECT(W)
*     .. Scalar Arguments ..
*
*     Logical function SELECT for use with ZGEESX (F08PPF)
*
*     Returns the value .TRUE. if the real part of the eigenvalue
*     W is positive.
*
      COMPLEX *16            W
*     .. Local Scalars ..
      LOGICAL                D
*     .. Intrinsic Functions ..
      INTRINSIC              DBLE
*     .. Executable Statements ..
      IF (DBLE(W).GT.0.0D0) THEN
         D = .TRUE.
      ELSE
         D = .FALSE.
      END IF
*
      SELECT = D
*
      RETURN
      END
```

9.2 Program Data

```
F08PPF Example Program Data

  4                                                                :Value of N

(-3.97, -5.04)  (-4.11,  3.70)  (-0.34,  1.01)  ( 1.29, -0.86)
( 0.34, -1.50)  ( 1.52, -0.43)  ( 1.88, -5.38)  ( 3.36,  0.65)
( 3.31, -3.85)  ( 2.50,  3.45)  ( 0.88, -1.08)  ( 0.64, -1.48)
(-1.10,  0.82)  ( 1.81, -1.59)  ( 3.25,  1.33)  ( 1.57, -3.44) :End of matrix A
```

9.3 Program Results

```
 F08PPF Example Program Results

 Number of eigenvalues for which SELECT is true =    2
 (dimension of invariant subspace)

 Schur matrix T
             1          2          3          4
 1      7.9982    -0.1954    -0.3852    -0.6995
       -0.9964    -1.0691     0.1986     0.1396

 2      0.0000     3.0023     0.7486     0.3468
        0.0000    -3.9998     0.5417     0.1036

 3      0.0000     0.0000    -6.0004    -0.4390
        0.0000     0.0000    -6.9998    -0.2558

 4      0.0000     0.0000     0.0000    -5.0000
        0.0000     0.0000     0.0000     2.0060

 Matrix of Schur vectors Z
```

```
          1       2       3       4
 1  -0.1730  0.0291  0.8532 -0.3510
     0.2669 -0.1912  0.0000  0.1013

 2   0.6924  0.0332 -0.0406 -0.4035
     0.0000  0.1789  0.3419  0.4540

 3   0.3324  0.0527  0.0756  0.6239
     0.4960 -0.4086  0.2807  0.0000

 4   0.2504  0.8716  0.0299 -0.0816
    -0.0147  0.0000 -0.2609 -0.3190

 Reciprocal of projection norm onto the invariant
 subspace for the selected eigenvalues
 RCONDE =  9.9E-01

 Reciprocal condition number for the invariant subspace
 RCONDV =  8.4E+00

 Approximate asymptotic error bound for selected eigenvalues   =  1.6E-15
 Approximate asymptotic error bound for the invariant subspace =  1.9E-16
```